
GPU Computing
Introduction and Motivation
Graphics processing units (GPUs) were created to offload graphical operations from the CPU.
Needed to determine pixel colour values for thousands of pixels;
the colour value of each pixel can be computed independently in parallel;
algorithms were simple enough and accelerated with hardware pipelines.
Over time, the GPUs became more capable.
Initially, a fixed sequence of operations was executed by various specialised processors
called shaders. These were executing predefined programs.
Shaders at each stage became programmable.
Eventually, a single compute unit would execute the entire pipeline, implemented in
software.

GPU Computing
Introduction and Motivation

This enabled GPUs to be used for suitable scientific algorithms.
At first, programs were written using APIs and languages dedicated to computer graphics.
GPU computing APIs and languages were subsequently developed (e.g. OpenCL and
CUDA).

Motivations for GPU Computing
Introduction and Motivation
Why using GPUs for scientific computing?
TFLOPS (SP)
Bandwidth (GB/s)

Intel Xeon E
0.46
39.5

Intel Xeon Platinum
2.27
131

NVIDIA A100
19.5
1638.4

AMD MI100
23.1
1228.8

Furthermore, GPUs are more power efficient and there is still room for performance growth
from one generation to the next.
Update: new generation of GPUs to be released:
AMD MI250x, with peak single precision (SP) performance of 47.9 TFLOPS, and
NVIDIA H100, with peak single precision performance of 48 TFLOPS.

GPU vs. CPU Architecture
Introduction and Motivation

Figure: A (very) simplified model of a CPU.
Figure: Representation of GPU architecture.

GPU vs. CPU Architecture
Introduction and Motivation

CPU:

GPU:

Fully-featured control logic for single
thread performance;

Large numbers of SIMD vector units for
throughput performance;

hierarchy of caches for low latency access
to data;

memory latency hidden via switching
instead of caching;

Increasing number of cores and Single
Instruction Multiple Data (SIMD) vector
unit length.

specialised hardware units (e.g. tensor
cores).

GPU Computing: Disadvantages
Introduction and Motivation

Existing CPU codes require (possibly extensive) modification to run on GPUs. This has
improved over time with libraries and compiler directives support.
Not all computations are well suited to the architecture.
Data transfers between host memory and GPU memory. Improved with:
GPUDirect: enables network and IO adapters to write to and read from GPU memory;
RDMA: provides direct communication between NVIDIA GPUs in remote systems;
NVLINK: fast NVIDIA interconnect between GPUs.

Further Considerations
Introduction and Motivation

CPUs are becoming more “GPU-like”:
increasing number of cores,
increasing vector parallelism.
GPUs are becoming more “CPU-like”:
increasing levels and sizes of cache,
increasing capability of thread management.
It is also possible to develop a single parallel code that can be executed on both CPUs and
GPUs with technologies such as OpenMP.

NVIDIA Tesla Architectures
Introduction and Motivation
Tesla is the NVIDIA’s line of products dedicated to HPC.
Architecture
Tesla
Fermi
Kepler
Maxwell
Pascal
Volta
Turing
Ampere

Capability
1.0 - 1.3
2.0
3.0 - 3.7
5.2
6.0 - 6.1
7.0
7.5
8.0

Models
C1060
C2050
K80
M60
P100
V100
T4
A100

New Features
Unified compute
ECC, 64-bit performance, caching
Dynamic parallelism, GPUDirect, Hyper-Q
Efficiency, increased shared memory
NVLink, HBM2, 16bit perf, page migration
Tensor cores
RT Cores
Multi-Instance GPU, AI-focused improvements

Major updates indicated with new architecture names, Compute Capability provides finer
grained versioning for the hardware architecture. GPU architecture is constantly improving
and evolving.

NVIDIA Tesla Architectures
Introduction and Motivation
Over the following decade, GPU performance has improved considerably.
Year
2007
2009
2011
2012
2013
2015
2016
2017
2018
2020

Architecture
Tesla
Tesla
Fermi
Kepler
Kepler
Maxwell
Pascal
Volta
Turing
Ampere

GHz
1.35
1.30
1.15
0.70
0.73
0.95
1.33
1.46
1.6
1.4

SMP
8x16
15x16
14x32
13x192
15x192
24x128
56x64
80x64
108x64

Cores
128
240
448
2496
2880
3072
3584
5120
2560
6912
x54

GFLOPs (FP64)
78
515
1175
1680
214
4670
7450
9700
x124

TDP W
171
188
247
225
235
250
250
300
70
400
x2.33

GFLOP/W
0.42
2.09
5.22
7.15
0.86
18.7
24.8
24.2
x57.8

NVIDIA Tesla Architectures
Introduction and Motivation
The memory capacity and bandwidth, as well as IO has also improved.
Year
2007
2009
2011
2012
2013
2015
2016
2017
2018
2020

Architecture
Tesla
Tesla
Fermi
Kepler
Kepler
Maxwell
Pascal
Volta
Turing
Ampere

Model
C870
C1060
C2050
K20
K40
M40
P100
V100
T4
A100

RAM (GB)
1.5
4
3
5
12
12
16
16
16
40
x26.6

RAM GB/s
76.8
102
144
208
288
288
732
900
320
1638
x21.3

IO GB/s
8
8
15.8
15.8
15.8
15.8
15.8
15.8
32
32
x4

GPU Programming Paradigms
Introduction and Motivation

Libraries. Optimised libraries for a wide range of common numerical algorithms. These
include libraries for linear algebra, statistics, signal and image processing, machine learning
and more.
Directives. Compiler directives provide acceleration while maintaining ability to run code
on CPU. GPU offloading is supported in both OpenACC and OpenMP.
Languages and APIs. Provides the most control and flexibility for writing GPU
applications. CUDA and OpenCL both support a wide range of parent languages, such as
C/C++, Fortran, Java, and Python.
One or more of these approaches may be appropriate depending on the application. Regardless
of the approach, an understanding of the GPU architecture is critical.

Using GPUs on Topaz
22 Volta GPU-enabled nodes, each with:
2x NVIDIA Tesla V100 GPUs (16GB memory, 5120
cores),
2x Intel Xeon 4215 CPUs (Cascade Lake, 16 cores),
196GB RAM,
100 Gbps Infiniband interconnect.
11 Pascal GPU-enabled nodes, each with:
4x NVIDIA Tesla P100 GPUs (16GB memory, 3584
cores, NVLINK),
2x Intel Xeon E5-2680 v4 CPUs (Broadwell, 14
cores),
128GB RAM.

V100 Nodes Configuration on Topaz
Using GPUs on Topaz

P100 Nodes Configuration on Topaz
Using GPUs on Topaz

Running GPU Jobs on Topaz
Using GPUs on Topaz

Log in using a ssh client to the login node of Topaz: topaz.pawsey.org.au.
Job submission is via the SLURM scheduler:
GPUs are in the gpuq partition (we are going to use only the V100 nodes),
Explicitly ask for gpu resources: --gres=gpu:<n>.

To compile CUDA code, we are going to use the NVIDIA HPC SDK package.

Compiling a CUDA Program
Using GPUs on Topaz
A batch script can be used to perform the compilation on a compute node:
1
2
3
4
5
6
7
8
9

#!/bin/bash --login
#SBATCH --nodes=1
#SBATCH --ntasks=1
#SBATCH --gres=gpu:1
#SBATCH --partition=gpuq
#SBATCH --time=00:05:00
#SBATCH --export=NONE
module use /group/courses0100/software/nvhpc/modulefiles
module load nvhpc/21.9
srun nvcc -o myprogram myprogram.cu # for CUDA C++
srun nvfortran -o myprogram myprogram.cuf # for CUDA Fortran
srun ./myprogram

However, for small programs an interactive session works just fine. nvcc is the compiler for
CUDA C++ programs, nvfortran for CUDA Fortran ones. All gpu code must be contained
in a .cu or .cuf file, respectively.

Exercise: Compile and Run a CUDA Program
Using GPUs on Topaz
1

log into Topaz: ssh <username>@topaz.pawsey.org.au

2

download script template from github:
git clone
https://github.com/PawseySC/gpu-programming-essentials.git

3

move into the query directory:
cd gpu-programming-essentials/query

4

write and submit a batch script to compile and execute the query program (wait for it to
finish)

5

remember that you can submit a job and check the status with the following commands:
sbatch --reservation=courses0100 <batchscript-file>
squeue -u <username>
cat slurm-<jobid>.out

Fixes for Common Problems
Using GPUs on Topaz

Avoid compiling on the login node. The login node may have a different architecture to
the GPU nodes. Use a batch job or an interactive session.
Do not run GPU programs on the login node, it does not have any GPUs.
Make sure the necessary module environment is loaded in submission scripts. Refer to the
Pawsey Documentation web pages for examples.
In interactive mode, executables must be run with srun in order for them to detect GPUs.

Introduction to CUDA

CUDA is a parallel computing framework developed by NVIDIA, which consists of:
Driver API: a low level interface for programming NVIDIA GPUs.
Runtime API: a higher level interface implemented on top of the driver API.
CUDA C++/Fortran language extensions to define GPU programs.
nvcc & nvfortran compiler drivers.
NSIGHT, cuda-gdb, cuda-memcheck: set of tools to analyse, debug and profile your
code.

Host and Device
Introduction to CUDA
The GPU is an accelerator that operates in conjunction with the CPU.
CPU is referred to as the host.
GPU is referred to as the device.
The host processor manages the operation of the device, typically:
Device memory allocations;
transfers of data between host and device memory;
submission of kernels to the device for execution;
synchronisation between host and device.
When writing a CUDA code, one actually defines two programs: the driver running on the CPU
and one or more kernels running on the GPU, concurrently and possibly in parallel.

Host and Device
Introduction to CUDA

Figure: The host issues commands to and synchronise with the GPU through the CUDA runtime.

CUDA Programming Model
Introduction to CUDA

Allows to develop application software that transparently scales to manycore GPUs with widely
varying numbers of cores. It does so through three key abstractions:
1

thread hierarchy,

2

memory hierarchy, and

3

barrier synchronisation.

Thread Hierarchy
Introduction to CUDA
Parallel computation is carried on by a grid of block threads.
Threads execute the same program (kernel).
Threads within a block can cooperate (i.e. exchange
information and synchronise). There can be a maximum
of 1024 threads in a block.
Each thread block executes independently of other
blocks. There can be a maximum of 2.2 billion blocks.
Grid can be 1D, 2D or 3D. Each thread has access to its
index within both the block and grid.
It reflects the type of decomposition a problem is expected to
have for a solution to be implemented efficiently (or even at
all).

Memory Hierarchy
Introduction to CUDA
There are different types of memory available of the GPU:
Global memory: large, GBs-sized DRAM accessible by all
threads in a grid. Accessible to host as well.
Shared memory: 64Kb-sized fast memory for each thread
block. Same on-chip memory used for L1 cache.
Local memory: thread-private memory for holding
automatic variables (slow as accessing global memory!).
Registers: there are 65k (vs. only dozens on CPU!)
registers available per block.
Other memories exist as well: constant and texture memories,
for specialised access patterns.

Vector Addition in C
Introduction to CUDA
Given two vectors of numbers A, B of length n, store the element-wise addition in a third
vector C . Each pair of respective elements in A and B can be added in parallel.
1

void vector_add(float *a, float *b, float *c, unsigned int n){
for(int i = 0; i < n; i++)
c[i] = a[i] + b[i];
}
int main(void){
unsigned int n = 100;
float *A = (float*) malloc(n * sizeof(float));
float *B = (float*) malloc(n * sizeof(float));
float *C = (float*) malloc(n * sizeof(float));
init_vec(A, n);
init_vec(B, n);
vector_add(A, B, C, n);
// do something with C ..
free(A); free(B); free(C);
return 0;
}

Vector Addition in Fortran
Introduction to CUDA

Vector addition in Fortran is so much simpler to implement!
1
2
3
4
5

8

program cpu_vector_add
implicit none
integer, parameter :: n = 100
real, dimension(n) :: a, b, c
call random_number(a)
call random_number(b)
b = a + b
end program cpu_vector_add

Vector Addition in CUDA C
Introduction to CUDA
1

__global__ void vector_add(float *a, float *b, float *c){
unsigned int i = threadIdx.x;
c[i] = a[i] + b[i];
}
int main(void){
unsigned int n = 100;
float *A = (float*) malloc(n * sizeof(float));
float *B = (float*) malloc(n * sizeof(float));
float *C = (float*) malloc(n * sizeof(float));
init_vec(A, n);
init_vec(B, n);
float *dev_A, *dev_B, *dev_C;
cudaMalloc(&dev_A, sizeof(float) * n);
cudaMalloc(&dev_B, sizeof(float) * n);
cudaMalloc(&dev_C, sizeof(float) * n);
// continue next slide ...

Vector Addition in CUDA C
Introduction to CUDA

// kernel shown again for convenience
__global__ void vector_add(float *a, float *b, float *c){
unsigned int i = threadIdx.x;
c[i] = a[i] + b[i];
}
...
cudaMemcpy(dev_A, A, sizeof(float) * n, cudaMemcpyHostToDevice);
cudaMemcpy(dev_B, B, sizeof(float) * n, cudaMemcpyHostToDevice);
vector_add<<<1, n>>>(dev_A, dev_B, dev_C);
cudaMemcpy(C, dev_C, sizeof(float) * n, cudaMemcpyDeviceToHost);
cudaDeviceSynchronize();
// Do something with C ...
free(A); free(B); free(C);
cudaFree(dev_A); cudaFree(dev_B); cudaFree(dev_C);
return 0;
}

Vector Addition in CUDA Fortran
Introduction to CUDA

1

! the following is the kernel
attributes(global) subroutine vector_add(A, B, C, n)
implicit none
integer, intent(in), value :: n
real, dimension(n), intent(in) :: A, B
real, dimension(n), intent(out) :: C
integer :: i
i = threadidx%x
C(i) = A(i) + B(i)
end subroutine vector_add

Vector Addition in CUDA Fortran
Introduction to CUDA

program gpu_vector_add
use cudafor
implicit none
integer, parameter :: n = 100
real, dimension(:), allocatable :: A, B, C
real, dimension(:), allocatable, device :: dev_A, dev_B, dev_C
integer :: istat
! allocate on main memory (CPU)
allocate(A(n), B(n), C(n))
! allocate on GPU memory
allocate(dev_A(n), dev_B(n), dev_C(n))
! initialize arrays on CPU
call random_number(A) ; call random_number(B)
! copy data on GPU
dev_A = A; istat = cudaMemcpy(dev_B, B, n)
call vector_add<<<1, n>>>(dev_A, dev_B, dev_C, n)
C = dev_C
istat = cudaDeviceSynchronize()
! do something with C...
deallocate (A, B, C, dev_A, dev_B, dev_C)
end program gpu_vector_add

Key takeaways
Introduction to CUDA

kernels are denoted with the __global__ keyword in C (attributes(global) in
Fortran) and do not have a return value. Thread index is available through the
struct/derived type threadIdx.
Cannot access CPU memory, must allocate memory on GPU and transfer data.
Some API calls are asynchronous, some are not. Use synchronisation methods to wait for
GPU to complete tasks.
Specify a kernel launch configuration (number of blocks and threads per block) using
triple angular brackets syntax.
Remember to deallocate memory both on CPU and GPU!

Error Handling
Introduction to CUDA
Due to performance constraints, runtime error handling is not as complex as on CPU. E.g. no
exceptions thrown when dividing by zero. Always check return values of CUDA runtime API
calls.
1

void __cuda_check_error(cudaError_t err, const char *file, int line){
if(err != cudaSuccess){
fprintf(stderr, "CUDA error (%s:%d): %s\n", file, line, cudaGetErrorString(err));
exit(1);
}
}
#define CUDA_CHECK_ERROR(X)({\
__cuda_check_error((X), __FILE__, __LINE__);\
})
...
CUDA_CHECK_ERROR(cudaMalloc(...));
CUDA_CHECK_ERROR(cudaMemcpy(...));
kernel<<<...>>>(...);
CUDA_CHECK_ERROR(cudaGetLastError());

Problem Decomposition
Introduction to CUDA
What if n > 1024? Use more than one thread block! How many?
1
2
3

nThreadsPerBlock = 1024;
nBlocks = (n + nThreadsPerBlock - 1) / nThreadsPerBlock;
vector_add<<<nBlocks, nThreadsPerBlock>>>(dev_A, dev_B, dev_C, n);

But now we may have more threads than data items. What can happen with the previously
defined kernel?

d1
t1

t2

block 1

t3

t4

block 2

t5

t6

block 3

t7

t8

block 4

Figure: Problem decomposition with n = 7 and 2 threads per block. The last thread is not mapped to
any data item.

Problem Decomposition
Introduction to CUDA
The way to fix the bug is to add an if statement in the code such that only threads with an
index corresponding to an array entry execute the summation. C and Fortran code:
1

__global__ void vector_add(float *a, float *b, float *c, unsigned int n){
unsigned int i = blockDim.x * blockIdx.x + threadIdx.x;
if(i < n){
c[i] = a[i] + b[i];
}
}
attributes(global) subroutine vector_add(A, B, C, n)
implicit none
integer, intent(in), value :: n
real, dimension(n), intent(in) :: A, B
real, dimension(n), intent(out) :: C
integer :: idx
idx = (blockidx%x - 1) * blockdim%x + threadidx%x
if (idx <= n) then
C(idx) = A(idx) + B(idx)
end if
end subroutine vector_add

Problem Decomposition
Introduction to CUDA

Switch from thread index within the block to index within the grid:
threadIdx.x (.y, .z): index of current thread within the block along the x (y , z) axis.
blockIdx.x (.y, .z): index of the block of current thread within the grid along the x
(y , z) axis.
blockDim.x (.y, .z): dimension of the block along the x (y , z) axis.

Problem Decomposition
Introduction To CUDA

In general,
1

Come up with a way of dividing your problem in small independent sub-problems.

2

Choose the number of threads per block. Max 1024 but limited also by other factors (e.g.
registers available per thread) and may depend on how you solve a sub-problem.

3

Compute the number of blocks needed to cover the entire input size.

This approach scales with the input size, as the programmer only thinks about solving a
sub-problem of fixed size and the code creates as many sub-problem instances as needed.

Exercise: Vector Addition for Arbitrary Vector Lengths
Introduction To CUDA

In this exercise, you will apply the suggestions presented on the previous slides to the CUDA
vector addition implementation present in the folder vector_add. In particular, you are
asked to modify the source code of gpu_vector_add.cu[f] such that it can compute the
addition of vectors of arbitrary length. For debugging purposes, try different vector sizes
(< 1024, = 1024, > 1024).

Exercise: Matrix Transpose
Introduction To CUDA

In this exercise you will write a CUDA program to compute the transpose of a matrix. Given a
matrix A of dimensions m × n, the transpose of A is a matrix A0 of dimensions n × m such that
A0ij = Aji ,
with 1 ≤ i ≤ n and 1 ≤ j ≤ m. Go to the folder matrix_transpose and complete the code
in matrix_transpose.cu[f].
Note: A is a 2D matrix but it is stored in a 1D vector. For each matrix element, you may need
to map its 1D coordinate to the corresponding 2D coordinate.

GPU Hardware Architecture

Figure: A schematic representation of the GPU architecture.

Streaming Multiprocessors
GPU Hardware Architecture

GPU comprises of several computing hardware units called Streaming Multiprocessors
(SM); they are the equivalent of a CPU core. On a datacenter GPU card there are on
average 80 SMs.
Each CUDA thread block is put in a queue, waiting to be scheduled on a SM for
execution.
All threads of a block are executed on the same SM.

The set of all SMs maps to a CUDA grid.
Streaming Multiprocessors access the global memory through the same L2 cache. Accesses
to global memory are coalesced into 128-byte wide memory transactions.

Inside a SM
GPU Hardware Architecture
Various specialised hardware units:
32-bit precision cores;
double precision cores (DPU);
tensor cores for multiply and add;
load and store units (LDST);
special function units (SFU), e.g. sin
and sqrt.

shared memory for fast communication
among threads.
warp scheduler decides which group of
threads to execute next.
dispatch unit submits the next instruction
to core pipeline for execution.

Single Instruction Multiple Threads
GPU Hardware Architecture

An instruction pipeline is a sequence of stages implemented in hardware to execute a set of
instructions.
A GPU instruction pipeline is able to execute the same instruction on 32 CUDA cores
(32bit ops, or 16 double precision units) at each step (i.e. at each clock).
Threads are executed in groups of 32 called warps.
Maximum efficiency when:
1
2

All threads in a warp follow the same execution path (i.e. same sequence of instructions),
and
there are enough instructions to be executed to hide latency.

Thread Hierarchy Revised
GPU Hardware Architecture

... 31

...
0

Figure: Each 32 consecutive threads in a block form a warp. The position of a thread within its warp is
called lane.

Thread Divergence
GPU Hardware Architecture

Thread Divergence
GPU Hardware Architecture

Threads in a warp can diverge, that is, follow different execution paths. This happens when
threads evaluate conditional statements differently.
execution paths are serialised; that is, executed one at a time.
threads that are being executed are called active.
inactive threads execute an artificial operation (no-op) that does nothing.
no sub-warp concurrency and synchronisation.
problem addressed with Independent Thread Scheduling from Volta architecture onwards
(advanced topic).

CPU vs GPU Parallelism
GPU Hardware Architecture
CPU Parallelism:
A CPU has a relatively small number of CPU cores capable of performing rather complex
operations in a serial fashion.
A CPU may have SIMD (single instruction multiple data) vector units.
A CPU core has a higher clock frequency than a SM of a GPU.
On A CPU, a thread executes a complex program and may use the whole SIMD vector
unit and all core registers.
GPU Parallelism:
A GPU has many SMs, each with several vector units (2 to 4) and circuits for special
operations.
Many instances of the same simple program running in parallel.
Each thread is mapped to an individual compute element in the vector unit.
Many lightweight threads share same set of registers on the SM.

Cooperating Threads

In many applications threads must cooperate to solve a problem. A more advanced example is
summing up all the elements of a vector.
Pn
c = i=1 A[i].
it is a special case of reduction, i.e. to compute a single (or few) items starting from many.
very popular algorithm featuring very important GPU programming patterns.
it will allow us to discuss thread synchronisation, shared memory, atomic functions.

Vector Reduction - Sequential CPU Implementation
Cooperating Threads
Sequential implementation on CPU, here is the relevant code. It will be our baseline. In our
examples, n is equal to one billion.

unsigned int n;
unsigned char *values = read_matrix(argv[1], &n);
struct timespec start, end;
clock_gettime(CLOCK_REALTIME, &start);
unsigned long long sum = 0;
for(unsigned int i = 0; i < n; i++)
sum += values[i];
clock_gettime(CLOCK_REALTIME, &end);
double time_spent = (end.tv_sec - start.tv_sec) + \
(end.tv_nsec - start.tv_nsec) / 1e9;

Execution time on Topaz: ≈ 2.14s

Vector Reduction Using CUDA - Host Code
Cooperating Threads
Allocate memory on device, set the output variable to zero, transfer data to GPU, launch
kernel.

unsigned long long sum = 0ull;
unsigned long long *dev_sum;
unsigned char *dev_values;
CUDA_CHECK_ERROR(cudaMalloc(&dev_values, sizeof(unsigned char) * nitems));
CUDA_CHECK_ERROR(cudaMalloc(&dev_sum, sizeof(unsigned long long)));
CUDA_CHECK_ERROR(cudaMemset(dev_sum, 0, sizeof(unsigned long long)));
CUDA_CHECK_ERROR(cudaMemcpy(dev_values, values,\
sizeof(unsigned char) * nitems, cudaMemcpyHostToDevice));
unsigned int nblocks = (nitems + NTHREADS - 1) / NTHREADS;
cudaEvent_t start, stop;
CUDA_CHECK_ERROR(cudaEventCreate(&start));
CUDA_CHECK_ERROR(cudaEventCreate(&stop));
CUDA_CHECK_ERROR(cudaEventRecord(start));
vector_reduction<<<nblocks, NTHREADS>>>(dev_values, nitems, dev_sum);
// continue next slide ...

Vector Reduction Using CUDA - Host Code Continued
Cooperating Threads

1

CUDA_CHECK_ERROR(cudaGetLastError());
CUDA_CHECK_ERROR(cudaEventRecord(stop));
CUDA_CHECK_ERROR(cudaDeviceSynchronize());
CUDA_CHECK_ERROR(cudaMemcpy(&sum, dev_sum, sizeof(unsigned long long),\
cudaMemcpyDeviceToHost));
CUDA_CHECK_ERROR(cudaDeviceSynchronize());
float time_spent;
CUDA_CHECK_ERROR(cudaEventElapsedTime(&time_spent, start, stop));

Use CUDA events to measure the execution time of the code.
976563 blocks are being created.

Vector Reduction Using CUDA - Kernel #1
Cooperating Threads

__global__ void vector_reduction(unsigned char *values, unsigned int nitems,
unsigned long long* result){
unsigned int idx = blockIdx.x * blockDim.x + threadIdx.x;
if(idx < nitems)
atomicAdd(result, values[idx]);
}

correspondence between threads and array items;
values are accumulated into one output global memory location;
atomic functions needed because of data race conditions due to concurrency;
atomic functions as a (limited) way for thread blocks to cooperate.
Execution time (kernel): 0.77 (only) 2.7× faster than serial version;
problem is all threads access the same memory location at the same time (furthermore
using an atomic function).

Atomic Functions
Cooperating Threads

An atomic function performs a read-modify-write atomic operation on one 32-bit or 64-bit
word residing in global or shared memory.
For example, atomicAdd() reads a word at some address in global or shared memory,
adds a number to it, and writes the result back to the same address.
The operation is atomic in the sense that it is guaranteed to be performed without
interference from other threads.
In other words, no other thread can access this address until the operation is complete.

Vector Reduction Using CUDA - Kernel #2
Cooperating Threads

__global__ void vector_reduction(unsigned char *values, unsigned int nitems,
unsigned long long* result){
unsigned int idx = blockIdx.x * blockDim.x + threadIdx.x;
__shared__ unsigned long long partial_sum;
if(threadIdx.x == 0) partial_sum = 0;
__syncthreads();
if(idx < nitems){
atomicAdd(&partial_sum, values[idx]);
}
__syncthreads();
if(threadIdx.x == 0) atomicAdd(result, partial_sum);
}

each thread block accumulates partial result in its shared memory;
thread zero of each block initialises shared memory and adds the block-wise partial result
to the final one;
blockwise synchronisation needed.
Execution time (kernel): 0.57s, 3.75× faster than serial version.

Block Synchronisation
Cooperating Threads
__syncthreads() waits until:
all threads in the thread block have reached this point and,
all global and shared memory accesses made by these threads prior to
__syncthreads() are visible to all threads in the block.
__syncthreads() is used to coordinate communication between the threads of the same
block and to avoid data races.
__syncthreads() is allowed in conditional code but only if the conditional evaluates
identically across the entire thread block, otherwise the code execution is likely to hang or
produce unintended side effects.
In practice, do not use it inside conditional blocks.

Vector Reduction Using CUDA - Kernel #3
Cooperating Threads
Same logic as previous one, but adding an intermediate warp-wide partial sum step.

__global__ void vector_reduction(unsigned char *values, unsigned int nitems,
unsigned long long* result){
unsigned int idx = blockIdx.x * blockDim.x + threadIdx.x;
__shared__ unsigned int partial_sums[WARP_SIZE];
unsigned int warpId = threadIdx.x / WARP_SIZE;
unsigned int laneId = threadIdx.x % WARP_SIZE;
if(laneId == 0) partial_sums[warpId] = 0;
__syncthreads();
if(idx < nitems)
atomicAdd(partial_sums + warpId, values[idx]);
__syncthreads();
if(laneId == 0 && warpId > 0)
atomicAdd(partial_sums, partial_sums[warpId]);
__syncthreads();
if(threadIdx.x == 0) atomicAdd(result, partial_sums[0]);
}

Thread Indices & Warps
Cooperating Threads

In the previous code we’ve made use of three different indexes for a thread:
1

idx: index of a thread within the grid (global index);

2

threadIdx.x: index of a thread within its block;

3

laneId: index of a thread within its warp.

Moreover, warpId is the index of the warp of a given thread.
Execution time (kernel): 0.015s, 142.6× faster than serial version.
To do better, we change algorithm to avoid atomic operations during warp reduction.

Exercise: Maximum Value in a Vector
Cooperating Threads

Sum is not the only function that can be applied during a reduction. In this exercise, you are
asked to write a CUDA program which returns the maximum value contained in a vector of
integers. Head to the folder vector_max and complete the code in vector_max.cu[f].

Parallel Reduction - Algorithmic Idea
Cooperating Threads

Exercise: Reduction Algorithm

It is important to get confident with the reduction algorithm shown in the previous slide, as it
is frequently used in many GPU programs. In this exercise, you are asked to write its serial
implementation on CPU. Head to the folder reduction_cpu and complete
reduction_cpu.[c|f90].

Vector Reduction Using CUDA - Kernel #4
Cooperating Threads

__global__ void vector_reduction(unsigned char *values, unsigned int nitems,
unsigned long long* result){
unsigned int idx = blockIdx.x * blockDim.x + threadIdx.x;
__shared__ unsigned int partial_sums[NTHREADS];
unsigned int warpId = threadIdx.x / WARP_SIZE;
unsigned int laneId = threadIdx.x % WARP_SIZE;
if(idx < nitems) partial_sums[threadIdx.x] = values[idx];
else partial_sums[threadIdx.x] = 0;
__syncthreads();
// step 1
for(unsigned int i = WARP_SIZE/2; i >= 1; i = i/2){
if(laneId < i){
partial_sums[threadIdx.x] += partial_sums[threadIdx.x + i];
}
__syncwarp();
}
__syncthreads();
// step 2 on the next slide...

Vector Reduction Using CUDA - Kernel #4 Continued
Cooperating Threads

// step 2
if(warpId == 0){
for(unsigned int i = WARP_SIZE/2; i >= 1; i = i/2){
if(laneId < i){
partial_sums[WARP_SIZE*laneId] += partial_sums[WARP_SIZE*(laneId + i)];
}
__syncwarp();
}
if(laneId == 0) atomicAdd(result, partial_sums[0]);
}
} // end of kernel

Vector Reduction Using CUDA - Kernel 4 Illustration
Cooperating Threads
Step 2.

Step 1.

Figure: In step 1, each warp accumulates its values at the location of its first thread in shared memory.
In Step 2, only the first warp is active, reducing the intermediate values to a single one at the very first
location in shared memory.

Vector Reduction Using CUDA - Kernel #4 Explanation
Cooperating Threads
1

shared memory size increased to the number of threads in a block;

2

each thread copies a vector element in shared memory.

3

Step 1: each warp in a block performs a parallel reduction on corresponding elements in
shared memory. Result is stored back in shared memory at the position corresponding to
the first thread of the warp. A total of 32 intermediate sums are computed.

4

Step 2: the first warp in each block performs a second reduction on the 32 warp-wise
partial sums to compute the block-wise partial sum.

Notice,
no atomic operation within a warp;
__syncwarp() is necessary starting from Volta architecture. Otherwise, some threads in
the warp could read outdated values if other threads in the same warp “are behind” with
execution.
Execution time: 0.012s, 178.3× faster than serial code.

Exercise: Maximum Value in a Vector Continued
Cooperating Threads

In light of the improvements made to the Vector Reduction code, you are now asked to apply
the parallel reduction algorithm to the vector_max.cu program.

Vector Reduction Using CUDA
Cooperating Threads

Can we do better? Yes, we can! Further improvements are explained at the end of the
presentation and left as optional (but strongly suggested!) material for the reader.

GPU Accelerated Libraries

Many scientific applications use similar algorithms to solve problems. GPU numerical libraries
provide optimised routines across a wide range of application areas, including:
Linear Algebra,
Signal and Image Processing,
Statistics,
Machine Learning.
They require no knowledge of GPU programming and just the use of libraries to accelerate
portions of your code may make full use of the GPU. However, they could cause unnecessary
host-device data transfer.

GPU Libraries: Linear Algebra
GPU Accelerated Libraries
Some NVIDIA libraries:
cuBLAS: standard Basic Linear Algebra Subroutines (BLAS);
nvBLAS :routes BLAS Level-3 routines to GPUs and CPUs;
cuSPARSE: basic linear algebra subroutines for sparse matrices;
cuSOLVER: factorisation, triangle, sparse least-squares, and eigenvalue solvers;
CHOLMOD: sparse Cholesky factorisation.
Third-party Libraries:
CULA Tools: dense and sparse matrix routines and solvers;
MAGMA: dense linear algebra library.

Example: Using cuBLAS
GPU Accelerated Libraries

#include <cublas_v2.h>
...
cublasStatus_t status;
// variable for cublas status
cublasHandle_t handle;
// variable for cublas handle
...
status = cublasCreate(&handle);
...
status = cublasSetVector(length, sizeof(float), hostArrayA, 1, deviceArrayA, 1);
status = cublasSetVector(length, sizeof(float), hostArrayB, 1, deviceArrayB, 1);
...
status = cublasSaxpy(handle, length, &c, deviceArrayA, 1, deviceArrayB, 1);
...
status = cublasGetVector(length, sizeof(float), deviceArrayB, 1, hostArrayB, 1);
...
status = cublasDestroy(handle);
...

Exercise: Using a GPU Library
GPU Accelerated Libraries

Move into the library subdirectory. The library.cu file creates two arrays, transfers
them to the GPU, adds them together, and transfers the result back to the host. Your task is
to add missing calls to cuBLAS to the code (hints at the bottom of the slide). When compiling
the code, remember to link cuBLAS library passing -lcublas to nvcc.
1

3

status = cublasSetVector(length,sizeof(float),hostArrayB, 1, deviceArrayB, 1);
status = cublasSaxpy(handle, length, &c, deviceArrayA, 1, deviceArrayB, 1);
status = cublasGetVector(length,sizeof(float),deviceArrayB, 1, hostArrayB, 1);

GPU Libraries: Signal and Image Processing
GPU Accelerated Libraries

NVIDIA Performance Primitives (npp): over 5000 primitives for image and signal
processing, including colour conversion, image compression, filtering, thresholding and
manipulation.
cuFFT: implements Fourier Transform. Similar routines to FFTW or MKL with batching
for parallelism;
FFmpeg: record, convert and stream audio and video content.

GPU Libraries: Statistics
GPU Accelerated Libraries

cuRAND: random number generators with various algorithms and distributions;
gpustats: python package with functionality similar to scipy.stats.

GPU Libraries: Machine Learning
GPU Accelerated Libraries

Libraries:
cuDNN: primitives for deep neural networks.
Frameworks:
Caffe: Python framework with pre-trained networks;
Chainer, Theano: Python framework;
PyTorch: Python framework based on Torch;
TensorFlow: Python/C++ framework.

GPU Libraries and Applications
GPU Accelerated Libraries

There are many more libraries available and hundreds of GPU-accelerated applications;
however, it is important to take the time to understand how the application is accelerated:
Which routines are GPU accelerated?
Are these routines used for your computations?

ADVANCED OPTIONAL MATERIAL

Warp Shuffle Functions
Advanced Cooperating Threads

The parallel reduction algorithm allows us to use warp shuffle functions. From the CUDA
Programming guide:
The __shfl_*_sync() intrinsics permit exchanging of a variable between threads
within a warp without use of shared memory. The exchange occurs simultaneously
for all active threads within the warp[...].
We are going to use __shfl_down_sync(), which copies from a lane with higher ID
relative to the caller.

Vector Reduction Using CUDA - Kernel #5
Advanced Cooperating Threads

__global__ void vector_reduction(unsigned char *values, unsigned int nitems,\
unsigned long long* result){
unsigned int idx = blockIdx.x * blockDim.x + threadIdx.x;
__shared__ unsigned int partial_sums[WARP_SIZE - 1];
unsigned int warpId = threadIdx.x / WARP_SIZE;
unsigned int laneId = threadIdx.x % WARP_SIZE;
unsigned int myvalue;
if(idx < nitems) myvalue = values[idx];
else myvalue = 0;
// step 1
for(unsigned int i = WARP_SIZE/2; i >= 1; i = i/2){
unsigned int up = __shfl_down_sync(0xffffffff, myvalue, i, WARP_SIZE);
if(laneId < i) myvalue += up;
}
if(laneId == 0 && warpId > 0) partial_sums[warpId-1] = myvalue;
__syncthreads();
// continues next slide...

Vector Reduction Using CUDA - Kernel #5 Continued
Advanced Cooperating Threads

// step 2
if(warpId == 0){
if(laneId > 0) myvalue = partial_sums[laneId-1];
for(unsigned int i = WARP_SIZE/2; i >= 1; i = i/2){
unsigned int up = __shfl_down_sync(0xffffffff, myvalue, i, WARP_SIZE);
if(laneId < i) myvalue += up;
}
if(laneId == 0) atomicAdd(result, myvalue);
}
} // end of kernel

Algorithm is the same, but using local variables instead of shared memory to perform the
reduction within a warp.
Shared memory is still used to exchange data among warps. In this case, warps 1 to 31
position their partial sums in shared memory for warp 0 to read. No need to allocate
shared memory for warp 0.
Execution time: 0.0075s; 1.5× faster than using shared memory; 285× faster than
sequential version.

Moving grid
Advanced Cooperating Threads
Creating many thread blocks causes many context switches which degrade performance.
1

Compute the number of blocks as a function of the number of SMs instead of input size.

2

But then, not all input may be covered by the grid of thread blocks.

3

Solution: let the grid move across the input. That is each thread block may cover more
than one portion of the input.

4

Another way of seeing it is that we are moving from a “one to one” mapping between
threads and vector elements to a “one to many” mapping.

d1
t1

t2

block 1

t3

t4

block 2

t1

t2

block 1

t3

t4

block 2

Vector Reduction Using CUDA - Kernel #6
Advanced Cooperating Threads

__global__ void vector_reduction(unsigned char *values, unsigned int nitems,
unsigned long long* result){
// ... not showing the usual definitions of idx, laneId, warpId and partial_sums
unsigned int gridSize = gridDim.x * blockDim.x;
unsigned int nloops = (nitems + gridSize - 1) / gridSize;
unsigned int l = 0, myvalue;
while(l++ < nloops){
if(idx < nitems) myvalue = values[idx];
else myvalue = 0;
for(unsigned int i = WARP_SIZE/2; i >= 1; i = i/2){
unsigned int up = __shfl_down_sync(0xffffffff, myvalue, i, WARP_SIZE);
if(laneId < i) myvalue += up;
}
if(laneId == 0 && warpId > 0) partial_sums[warpId - 1] = myvalue;
__syncthreads();
// continue next slide...

Vector Reduction Using CUDA - Kernel #6 Continued
Advanced Cooperating Threads

...
if(warpId == 0){
if(laneId > 0) myvalue = partial_sums[laneId - 1];
for(unsigned int i = WARP_SIZE/2; i >= 1; i = i/2){
unsigned int up = __shfl_down_sync(0xffffffff, myvalue, i, WARP_SIZE);
if(laneId < i) myvalue += up;
}
if(laneId == 0) atomicAdd(result, myvalue);
}
idx += gridSize;
__syncthreads();
}
}

160 appears to be the optimal number of blocks (2× the number of SMs on a V100).
Execution time: 0.0053s; 403.7× faster than serial version.

Vector Reduction Using CUDA - Kernel #7
Advanced Cooperating Threads
We don’t have to move threads in a block at the same time. In this variation, we let all threads
accumulate the corresponding input values first (lines 7 − 8), then perform the reduction
(which now happens only one time per warp!).

__global__ void vector_reduction(unsigned char *values, unsigned int nitems,
unsigned long long* result){
// ... not showing the usual definitions of idx, laneId, warpId and partial_sums
unsigned int gridSize = gridDim.x * blockDim.x;
unsigned int nloops = (nitems + gridSize - 1) / gridSize;
unsigned int l = 0, myvalue;
for(unsigned int l = 0; l < nloops; l++, idx+=gridSize)
if(idx < nitems) myvalue += values[idx];
for(unsigned int i = WARP_SIZE/2; i >= 1; i=i/2){
unsigned int up = __shfl_down_sync(0xffffffff, myvalue, i, WARP_SIZE);
if(laneId < i) myvalue += up;
}
if(laneId == 0 && warpId > 0) partial_sums[warpId] = myvalue;
//continue next slide...

Vector Reduction Using CUDA - Kernel #7 Continued
Advanced Cooperating Threads

__syncthreads();
if(warpId == 0){
if(laneId > 0) myvalue = partial_sums[laneId - 1];
for(unsigned int i = WARP_SIZE/2; i >= 1; i = i/2){
unsigned int up = __shfl_down_sync(0xffffffff, myvalue, i, WARP_SIZE);
if(laneId < i) myvalue += up;
}
if(laneId == 0) atomicAdd(result, myvalue);
}
}

Execution time: 0.002s, 1070× faster than serial version.

References and Additional Readings

Content covered in this lecture and more can be found on https://developer.nvidia.com/.
Search for the CUDA Programming Guide to start with.

Thank You.

